1.5.1 Singulární body

```
> restart;
> with(LinearAlgebra):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Rovnice kvadriky:

```
> Kv:=expand(X.K.Transpose(X))=0;
```

$$Kv := x^2\, a(1,1) + 2\, x\, y\, a(1,2) + 2\, x\, z\, a(1,3) + 2\, x\, a(1,4) + y^2\, a(2,2) + 2\, y\, z\, a(2,3) + 2\, y\, a(2,4) + z^2\, a(3,3) + 2\, z\, a(3,4) + a(4,4) = 0$$

Bod přímky:

```
> M:=[m,n,p];
```

$$M := [m, n, p]$$

Parametrické rovnice přímky:

```
> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Primka := [x = m + t\, u, y = n + t\, v, z = p + t\, w]$$

Parametrické rovnice přímky dosazené do rovnice kvadriky:

```
> Kv1:=simplify(eval(Kv,Primka));
```

$$Kv1 := 2\, a(1,1)\, m\, t\, u + 2\, a(1,2)\, m\, t\, v + 2\, a(1,2)\, t\, u\, n + 2\, a(1,2)\, t^2\, u\, v + 2\, a(1,3)\, m\, t\, w + 2\, a(1,3)\, t\, u\, p + 2\, a(1,3)\, t^2\, u\, w + 2\, a(2,2)\, n\, t\, v + 2\, a(2,3)\, n\, t\, w + 2\, a(2,3)\, t\, v\, p + 2\, a(2,3)\, t^2\, v\, w + 2\, a(3,3)\, p\, t\, w + a(1,1)\, t^2\, u^2 + 2\, a(1,2)\, m\, n + 2\, a(1,3)\, m\, p + 2\, a(1,4)\, t\, u + a(2,2)\, t^2\, v^2 + 2\, a(2,3)\, n\, p + 2\, a(2,4)\, t\, v + a(3,3)\, t^2\, w^2 + 2\, a(3,4)\, t\, w + a(4,4) + 2\, a(3,4)\, p + a(1,1)\, m^2 + 2\, a(1,4)\, m + a(2,2)\, n^2 + 2\, a(2,4)\, n + a(3,3)\, p^2 = 0$$

Koeficient B rovnice $At^2 + Bt + C = 0$ společných bodů kvadriky a přímky:

```
> B:=coeff(lhs(Kv1),t)/2;
```

$$B := a(1,1)\, m\, u + a(1,2)\, m\, v + a(1,2)\, u\, n + a(1,3)\, m\, w + a(1,3)\, u\, p + a(2,2)\, n\, v + a(2,3)\, n\, w + a(2,3)\, v\, p + a(3,3)\, p\, w + a(1,4)\, u + a(2,4)\, v + a(3,4)\, w$$

```
> B:=collect(B,[u,v,w]);
```

$$B := (a(1,1)\, m + a(1,3)\, p + a(1,4) + a(1,2)\, n)\, u$$

$$+ (a(1,2)\,m + a(2,2)\,n + a(2,4) + a(2,3)\,p)\,v$$
$$+ (a(3,3)\,p + a(2,3)\,n + a(1,3)\,m + a(3,4))\,w$$

Soustava rovnic - podmínek singularity bodu (bodů) $M = [m, n, p]$:

```
> r1:=sort(coeff(B,u),[m,n,p])=0; r2:=sort(coeff(B,v),[m,n,p])=0;
  r3:=sort(coeff(B,w),[m,n,p])=0;
  r4:=sort(simplify(coeff(lhs(Kv1),t,0)-m*lhs(r1)-n*lhs(r2)-p*lhs(
  r3)),[m,n,p])=0;
```

$$r1 := a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4) = 0$$
$$r2 := a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4) = 0$$
$$r3 := a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4) = 0$$
$$r4 := a(1,4)\,m + a(2,4)\,n + a(3,4)\,p + a(4,4) = 0$$

```
> K;
```

$$\begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

```
> Delta:=Determinant(K):
```

Příklad: Volme $M = [0, 0, 0]$. Hledáme kvadriku s tímto singulárním bodem.

```
> M:=[0,0,0]:
> res1:=solve(eval({r1,r2,r3,r4},[m=0,n=0,p=0]),{a(1,4),a(2,4),a(3
  ,4),a(4,4)});
```

$$res1 := \{a(1,4) = 0,\ a(2,4) = 0,\ a(3,4) = 0,\ a(4,4) = 0\}$$

```
> assign(res1);
> Kv;
```

$$x^2\,a(1,1) + 2\,x\,y\,a(1,2) + 2\,x\,z\,a(1,3) + y^2\,a(2,2) + 2\,y\,z\,a(2,3) + z^2\,a(3,3) = 0$$

```
> Kv1:=x^2+y^2-z^2=0;
```

$$Kv1 := x^2 + y^2 - z^2 = 0$$

```
> Stred:=[0,0,0]:
```

Pro grafické znázornění kvadriky použijeme její parametrické vyjádření:

```
> Kv1p:=subs(x=t*cos(u),y=t*sin(u),Kv1);
```

$$Kv1p := t^2 \cos(u)^2 + t^2 \sin(u)^2 - z^2 = 0$$

```
> Res_z:=solve(Kv1p,{z});
```

$$Res\_z := \{z = t\},\ \{z = -t\}$$

Stačí uvažovat jenom jedno z řešení pro t:

```
> Kv1p[1]:=eval([t*cos(u),t*sin(u),z],Res_z[1]);
  Kv1p[2]:=eval([t*cos(u),t*sin(u),z],Res_z[2]);
```

$$Kv1p_1 := [t\cos(u), t\sin(u), t]$$

$$Kv1p_2 := [t\cos(u), t\sin(u), -t]$$

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> Plocha:=plot3d({Kv1p[1],Kv1p[2]},t=0..5,u=0..2*Pi,orientation=[52,63],grid=[50,50],style=patchcontour,axes=frame,color=COLOR(RGB,250/255,250/255,250/255),light=[90,-5,1,1,1],tickmarks=[3,3,3],scaling=constrained):
> Bod:=plottools[sphere](Stred,0.08,color=black):
> Primka1:=plot3d(evalm(Stred+t*([-1,3,2])),t=-2..2,s=-1..1,thickness=2):
> Primka2:=plot3d(evalm(Stred+t*([1,1.5,3])),t=-2.1..2.2,s=-1..1,thickness=2):
> Primka3:=plot3d(evalm(Stred+t*([3,-2.5,1.6])),t=-2..2,s=-1..1,thickness=2):
> plots[display](Plocha,Bod,Primka1,Primka2,Primka3,orientation=[-152,71]);
```




```
>
>
```